Maple

Properties
- Software package
- implemented in the programing language C
- available for many Operating Systems, e.g. Windows, Unix, Linux
- desined for numerical and **symbolic** expressions

- includes untilities for algebra, calculus, discrete mathematics, graphics, ...

History
- 1980: first development at the University of Waterloo, Canada
- 1988: Waterloo Maple Software was founded in order to sell and improve the software
- currently: version 12

$$8 \tag{1}$$

Getting started
- login to one of the machines in the pool in the Piloty building
- open a shell / a terminal
- type: xmaple (or maple, if you would like to work without windows; e.g. remote from home)

Menu bar at the top:
- allows you to save or load and edit your maple session
  e.g. clicking on the ***File*** menu and selecting ***Save*** allows to save the current worksheet
- below the menu bar, there is a collection of shortcut-buttons

Maple Help
- help menu, "Maple Help"
- ?command; e.g. ?solve, if you know the keyword in advance

$$plot\left(\cos\left(\frac{x}{2}\right) + \sin(2\,x),\, x = 0\,..4\,\pi\right);$$



$plot3d\left(1.3^{x}\sin(y), x = -1..2\pi, y = 0..\pi, coords = spherical, style = patch\right);$



- the help-window has two panels: the Help Navigator on the left and the help itself on the right
- each help page contains some examples; copying an example and pasting it into the worksheet is possible

Content

Basic Conventions

Entering a command, example

```
> restart;
>
```

Arithmetic operators

Addition	+	3 + 4
Substraction	-	$x - y$
Multiplication	*	2*x
Division	/	x / y
Exponentiation	^	3^4
Factorial	!	3 !

The precedence order follows the mathematical conventions:

```
> 56 − 4·2;
                                48                                    (2)
> (56 − 4)·2;
                                104                                   (3)
```

Special commands to access previous results
% latest one
%% last but second command
%%% last but third command

```
> #this is a comment
> 2·4; # most recent result becomes 8
                                8                                     (4)
> %· 12.4; # this computes 8·12.4. 99.2 becomes most recent result
                                99.2                                  (5)
> %%− %; # computes 8-99.2
                                -91.2                                 (6)
>
```

Defining Expressions with ":="
- expression: combination of numbers, variables and operators

- Syntax is *name:=expression*
- maybe most used concept in Maple
- notice the difference between an expression and a function:

Example

> $f := x^2 - 3\cdot x + 13;$

$$f := x^2 - 3\,x + 13 \tag{7}$$

> $g := x \to 3\cdot x - 3;$

$$g := x \to 3\,x - 3 \tag{8}$$

> $h(x) := 3\cdot x - 4;$

$$h := x \to 3\,x - 4 \tag{9}$$

> $plot([f, h(x), g(x)], x = 1..10);$



>

If you make a mistake, you can go back with the cursor, change the command-line and re-execute the line.

Basic Data Structures

- fundamental data structures: expression sequences, lists, sets. (e.g. used as parameters in maple commands)

Sequences, implicitely or with command seq(f(i),i=m..n)

> $3, 5, x, 4;$

$$3, 5, x, 4 \quad \textbf{(10)}$$

> $s := 3, 5, x, 4;$

$$s := 3, 5, x, 4 \quad \textbf{(11)}$$

> $evalf(\pi);$

$$3.141592654 \quad \textbf{(12)}$$

> $t := seq(i^2, i = 2..5);$

$$t := 4, 9, 16, 25 \quad \textbf{(13)}$$

> $t2 := 3, t;$

$$t2 := 3, 4, 9, 16, 25 \quad \textbf{(14)}$$

A list
- is an expression sequence enclosed in square brackets
- preserves order and repetition of elements

A set
- is an expression sequence enclosed in curly brackets
- does not preserve order an does not contain the same element several times

> $list1 := [5, 4, 3, 5, 4, 3];$

$$list1 := [5, 4, 3, 5, 4, 3] \quad \textbf{(15)}$$

> $list2 := [3, 4, 5];$

$$list2 := [3, 4, 5] \quad \textbf{(16)}$$

> $set1 := \{5, 4, 3, 5, 4, 3\};$

$$set1 := \{3, 4, 5\} \quad \textbf{(17)}$$

> $set2 := \{4, 5, 3\};$

$$set2 := \{3, 4, 5\} \quad \textbf{(18)}$$

> $s := [op(list2), op(list2)];$

$$s := [3, 4, 5, 3, 4, 5] \quad \textbf{(19)}$$

Numerical Computation

Fraction numbers and floating point numbers

- fractions are not reduced to floating point approximations
- exact computations with fractions
- with *evalf*, the fraction can be converted to a floatring point number with D*igits* many digits.

> $x := \frac{9}{8} + \frac{6}{5};$

$$x := \frac{93}{40} \tag{20}$$

> $evalf(\%);$

$$2.325000000 \tag{21}$$

> $evalf(x);$

$$2.325000000 \tag{22}$$

> $Digits := 20;$

$$Digits := 20 \tag{23}$$

> $evalf(x);$

$$2.3250000000000000000 \tag{24}$$

> $\frac{9}{8.0} + \frac{6}{5};$ *# a floating number in the expression leads to implicit evalf*

$$2.3250000000000000000 \tag{25}$$

>

Integer numbers

- arbitrary large integers (as far as there is enough memory)

> 10000!;

28462596809170545189064132121198688901480514017027992307941799942744113400037\ **(26)**

6444377299078675778477 5[...35460 digits...]

0000000000000000000000000000000000000000000000000000000000000000000000000000

0000000000000000000000000

> $Digits := 10;$ 100.0!;

$$Digits := 10$$

$$9.332621544\ 10^{157} \tag{27}$$

>

Complex Numbers

- a complex number z is of the form a + bi, with i^2= -1 and a,b ∈ℝ. a = Re(z) is the realpart of z and b=Im(z)
  is the imaginary part of z
- two complex numbers are equal if and only if their real parts and their imaginary parts are equal

- Complex numbers are added, subtracted, multiplied, and divided by formally applying the associative,
  commutative and distributive laws of algebra, together with the equation i 2 = -1.
  Addition : (a+bi) + (c+di) = (a+c) + (b+d)i
  Substraction : (a+bi) - (c+di) = (a-c) + (b-d)i
  Multiplication: $(a + bi) \cdot (c + di) = (ac - bd) + (bc + ad)i$

Division $\quad : \dfrac{a+bi}{c+di} = \dfrac{ac+bd}{c^2+d^2} + \dfrac{bc-ad}{c^2+d^2}i$, with c or d not equal to 0

- with the given definitions of addition, substraction, multiplication, division, and
  the additive identity (zero-element) 0 + 0i,
  the multiplicative identity (one-element) 1 + 0i,
  the addidive inverse of a number a + bi: -a - bi, and
  the multiplicative inverse of a + bi: $\dfrac{a}{a^2+b^2} + \dfrac{-b}{a^2+b^2}$,
  the complex numbers $\mathbb{C}$ are a *field* (dt: Körper)

> $\dfrac{(3+3 \cdot I)}{(2+6 \cdot I)}$;

$$\frac{3}{5} - \frac{3}{10}\,\mathrm{I} \tag{28}$$

> $\left(\dfrac{3}{3^2+5^2} + \dfrac{(-5)}{3^2+5^2} \cdot I\right) \cdot (3+5 \cdot I)$;

$$1 \tag{29}$$

Symbolic Computations

> $c := \left(\dfrac{a}{a^2+b^2} + \dfrac{-b}{a^2+b^2} \cdot I\right) \cdot (a+b \cdot I)$;

$$c := \left(\frac{a}{a^2+b^2} - \frac{\mathrm{I}\,b}{a^2+b^2}\right)(a+\mathrm{I}\,b) \tag{30}$$

>

>

Simplifying an Expression

Maple knows many functions for symbolic expression computations. Here, the most commonly used ones.

The simplify command tries to find a simpler equivalent for a given expression. The rules for the simplification steps follow some heuristics (but of course, the chosen simplification steps themselves are correct).

> *simplify*(%);

$$-\frac{-a^2-b^2}{a^2+b^2} \tag{31}$$

> *simplify*(%);

$$1 \tag{32}$$

The following expression leads to a surprising answer. Why? Because somewhere above, we already defined x. Thus: be careful and alert!

> $simplify(\sin(x)^2 \cdot x^4 + \cos(x)^2 \cdot x^4);$

$$\frac{74805201}{2560000} \quad (33)$$

> $simplify(\sin(y)^2 \cdot y^4 + \cos(y)^2 \cdot y^4);$

$$y^4 \quad (34)$$

> $restart;$

> $simplify(\sin(x)^2 \cdot x^4 + \cos(x)^2 \cdot x^4);$

$$x^4 \quad (35)$$

>

>

Expanding a Polynomial

The *expand* command produces a sum of products for polynomials.
A polynomial is a mathematical expression consisting of a sum of terms each of which is a product of a constant and one or more variables with non-negative integral powers. If there is only a single variable, x,
the general form is given by $a_0 x^n + a_1 x^{n-1} + a_2 x^{n-2} + \ldots + a_{n-1} x + a_n$, where the a_i are constants (called coefficients).

Examples:

> $restart; p := (x+3) \cdot (x-7);$

$$p := (x+3)\,(x-7) \quad (36)$$

> $expand(p);$

$$x^2 - 4x - 21 \quad (37)$$

> $q := (x+3) \cdot (x-7) \cdot (x+7); r := (x+25) \cdot (x-7) \cdot (x+9);$
$expand\left(simplify\left(expand\left(\frac{q}{r}\right)\right)\right);$

$$q := (x+3)\,(x-7)\,(x+7)$$

$$r := (x+25)\,(x-7)\,(x+9)$$

$$\frac{x^2}{(x+25)\,(x+9)} + \frac{10\,x}{(x+25)\,(x+9)} + \frac{21}{(x+25)\,(x+9)} \quad (38)$$

Factorize a Polynomial Expression

The command *factor* is the opposite of the*expand* command. It factorizes polynomial exprressions.

> $factor(x^2 - 1);$

$$(x-1)(x+1) \tag{39}$$

> $factor(\%\%);$

$$\frac{(x+3)(x+7)}{(x+25)(x+9)} \tag{40}$$

Normalize fractions

Restructures rational expressions. If possible, an expression is converted to factored normal form. This is the form numerator/denominator, where the numerator and denominator are relatively prime polynomials with integer coefficients.
I.e., common factors are canceled.

> $normal\left(\frac{x^5}{x+1} + \frac{x^4}{x+1}\right);$

$$x^4 \tag{41}$$

> $normal\left(\frac{1}{x} + \frac{x}{x+1}\right);$

$$\frac{x^2 + x + 1}{x(x+1)} \tag{42}$$

> $normal\left(\frac{1}{x} + \frac{x}{x+1}, expanded\right);$

$$\frac{x^2 + x + 1}{x^2 + x} \tag{43}$$

> $simplify\left(\frac{x^5}{x+1} + \frac{x^4}{x+1}\right);$

$$x^4 \tag{44}$$

> $normal\left(\frac{q}{r}\right);$ *#in the output are nominator and denominator relatively prime.*

$$\frac{(x+3)(x+7)}{(x+25)(x+9)} \tag{45}$$

> $normal\left(\frac{q}{r}, expanded\right);$

$$\frac{x^2 + 10x + 21}{x^2 + 34x + 225} \tag{46}$$

Programming with Maple

Simple commands

e.g. all direct commands we saw so far.

Comparison Operators (<, >, >, <=, >=)

> $a := 0; b := 1;$

$$a := 0$$
$$b := 1 \tag{47}$$

> $evalb(a = 0);$ *#evalb prints boolean results to screen*

$$true \tag{48}$$

> $evalb(b > 2);$

$$false \tag{49}$$

> $evalb(b + a \le 0);$

> $a := 0;$

$$a := 0 \tag{50}$$

Flow Control (if, for, while, ...)

if <conditional expression> **then** <statement sequence>
| **elif** <conditional expression> **then** <statement sequence> |
| **else** <statement sequence> |
end if

(Note: Phrases located between | | are optional.)

> **if** $(a > 0)$ **then** $f := x^2$ **fi**;

> **if** $(a = 0)$ **then** $f := x^2$ **fi**;

$$f := x^2 \tag{51}$$

>

> **if** $(a < 9)$ **then**
$f := x^2 + 1;$ # *";" is necessary, because: several statements without structure*
$g := x^2$ # *";" not necessary*
else
$g := x^2 + 1;$
$f := x^2$
end if;

$$f := x^2 + 1$$
$$g := x^2 \tag{52}$$

The **for ...while ... do** loop

>

>

1) Print even numbers from 6 to 10.

```
> for i from 6 by 2 to 10 do print(i) end do;
```

$$6$$

$$8$$

$$10 \tag{53}$$

2) Find the sum of all two-digit odd numbers from 11 to 99.

```
> mysum := 0 :
  for i from 11 by 2 while i < 100 do
     mysum := mysum + i;
     #print(mysum);
   end do: #a ; instead a : leads to different outputs
   mysum;
```

$$2475 \tag{54}$$